
EQUACIONS DIFERENCIALS

Aquest tema conté essencialment dues parts: la primera consisteix en un repàs de càlcul de funcions d’una
variable on s’introdueixen les eines necessàries que ens proporciona el Maple. La segona part consisteix en
l’estudi d’equacions diferencials: significat i resolució en certs casos.

1 Repàs de càlcul

En aquesta primera part del tema 3, veurem com es realitzen les operacions bàsiques amb funcions com la
derivació, la integració i la representació gràfica de funcions d’una variable. Aquesta secció ens servirà de
repàs per entendre millor la resolució i interpretació d’equacions diferencials que és la segona part d’aquest
tema.

Abans de començar recordeu que és recomanable començar la sessió amb la comanda restart que neteja
la memòria de la sessió de Maple, tant de variables com de paquets o llibreries.

> restart;

1.1 Funcions

El primer que hem de fer és recordar com es defineixen funcions amb Maple, per això cal que utlitzem
l’operador -> en què indiquem quina és la variable de la funció. D’aquesta manera és molt senzill avaluar

les funcions tant en valors concrets com en paràmetres.
> f:=x->2*x-3;
> f(2);
> f(-3);
> f(t);
> f:=x->2*x-a;
> f(-1);
> f(t);

També ens permet definir funcions que depenen de més d’una variable i avaluar-les.
> g:=(x,y)->2*x-3*y;
> g(2,-1);
> g(a,b);

1.2 Càlcul de derivades

La comanda bàsica de Maple pel càlcul de derivades d’expressions respecte alguna variable és la comanda
diff . Hem d’especificar primer la funció que volem derivar i després la variable respecte la qual derivem,
és a dir, escrivim diff(f(x),x). Recordeu també que els resultats obtinguts es poden simplificar amb la
comanda simplify . Observeu el seu funcionament en els següents exemples.

Exemples
> restart;

1. Calculem la derivada de e−x2
.

> diff(exp(-x^2),x);

2. Una altra funció,

> diff(arctan(ln(sin(x)/exp(x))),x);

> simplify(%);

3. El Maple aplica les regles de derivació de manera formal, en aquest exemple ens diu com calcular la
derivada d’un producte de funcions qualsevol i d’una exponencial de funcions.

> diff(f(x)*g(x),x);

> diff(f(x)^g(x),x);

Observeu que a la comanda diff li passem una expressió i ens retorna una altra expressió. Si volem
que el resultat sigui una funció també, hem d’utilitzar la comanda D . Li passem una funció i ens retorna

una funció. Què fa l’última comanda del següent exemple?

Exemples
> f:=x->sqrt(x^3);
> D(f);
> D(f)(2);

En el cas en què una funció depèn de vàries variables, aleshores es pot definir de manera semblant la derivació
respecte alguna d’aquestes variables, i a més, podem calcular les derivades parcials respecte cadascuna de les
variables fent diff(f(x,y),x) o diff(f(x,y),y). Amb la comanda D, no li hem d’indicar la variable sinó
la posició (o ordre) que ocupa la variable en la definició de la funció, entre corxets. Per exemple,

> f:=(x,y,z)->sqrt(x^2+y^2+z^2);
> diff(f(x,y,z),x);
> diff(f(x,y,z),y);
> diff(f(x,y,z),z);
> D[1](f);
> D[2](f);
> D[3](f);

1.2.1 Derivades successives

Per a calcular la derivada segona d’una funció sempre podem aplicar la comanda diff iterant dues vegades.
Per exemple, per la funció logaritme neperià podem calcular la derivada segona i tercera de la següent forma,

> diff(ln(x),x); diff(%,x);diff(%,x);

Tot i aix́ı, hi ha una manera més senzilla de fer-ho. Quan especifiquem la variable, podem escriure $n
on n és el número de vegades que volem derivar la funció. En el següent exemple veureu com calcular la
derivada segona i tercera de la funció logaritme neperià ln(x).

> diff(ln(x),x$1);
> diff(ln(x),x$2);
> diff(ln(x),x$3);
> f:=x->ln(x); D[1$2](f); D[1$3](f);

Observeu que si passa si escriviu x$4 a una ĺınia de comandes,

> x$4;

Si la funció depèn de vàries variables, podem fer derivades succesives alternant les variables. Per exemple,
> diff(ln(x+y+z),x);diff(%,y);diff(%,z);
> diff(ln(x+y+z),x,y,z);

Exercicis

1. Calculeu les derivades de les funcions f(x) = xx, g(x) = cos(x2), f(x) + g(x), f(x)g(x) i
f(x)
g(x)

.

2. Calculeu les derivades parcials de f(x, y, z) = ex/y + ez/y respecte x, y i z.

3. Calculeu la derivada segona i la derivada cinquena de la funció f(x) = x5ex.

1.3 Recta tangent i pla normal

La derivada d’una funció f(x) en un punt té un significat geomètric com el pendent de la recta tangent a la
gràfica de f(x) en aquell punt. La recta tangent a f(x) en el punt x = a s’escriu com y = f ′(a)(x− a) + f(a).

Exemple 1

Calcularem i representarem gràficament la recta tangent a la funció f(x) = ex + sin(x) + x en el punt x = 0.
Comencem definint la funció (recordeu que per definir una funció fem servir l’expressió ->),

> restart;

> f:=x->exp(x)+sin(x)+x;

En una variable anomenada der guardarem el resultat de derivar la funció.

> der:=diff(f(x),x);

El pendent de la recta tangent s’obté avaluant la derivada en el punt x = 0.

> subs(x=0,der);

> m:=evalf(%);

Noteu que dóna el mateix que utilitzar la comanda D que ens retorna una funció que podem evaluar
directament.

> D(f)(0);

Definim la recta tangent seguint la fórmula.

> r:=x->m*(x-0)+f(0);

I finalment observem el resultat fent una gràfica conjunta de la funció i la recta amb la comanda plot .

> plot([f(x),r(x)],x=-2..2);

Donada una corba en forma paràmetrica, també podem determinar la seva recta tangent i el pla normal.
Sigui l’expressió de la corba (x(t), y(t), z(t)) a R3 de forma paramètrica. Aleshores fixat un punt de la corba
P = (x(t0), y(t0), z(t0)) que correspon al paràmetre t0, el vector tangent s’obté derivant respecte t cadascuna
de les coordenades i després substituint, v = (x′(t0), y′(t0), z′(t0)). Per tant, la recta tangent en forma
paramètrica és

(x(t), y(t), z(t)) = (x0 + tx′(t0), y0 + ty′(t0), z0 + tz′(t0))

i el pla normal és

x′(t0)(x− x0) + y′(t0)(y − y0) + z′(t0)(z − z0) = 0.

Exemple 2

Anem a calcular i dibuixar la recta tangent i el pla normal a la corba x(t) = t sin(t), y(t) = cos(t) i z(t) = t
en el punt P = (π/2, 0, π/2) que s’obté al fer t = π

2 . Primer dibuixem la corba, recordeu que les comandes
gràfiques es troben al paquet plots i que les corbes paramètriques es dibuixen amb spacecurve . Un pla

es pot dibuixar a partir de l’equació cartesiana amb la comanda implicitplot3d

> restart;
> x:=t->t*sin(t); y:=t->cos(t); z:=t->t;
> with(plots):
> spacecurve([x(t),y(t),z(t)],t=0..2*Pi, axes=normal);

El vector tangent a un punt qualsevol s’obté derivant i després substituim al punt concret t = π/2 per a
poder obtenir les tres coordenades del vector tangent.

> v:=vector([diff(x(t),t), diff(y(t),t), diff(z(t),t)]);
> v1:=subs(t=Pi/2,diff(x(t),t));
> v2:=subs(t=Pi/2,diff(y(t),t));
> v3:=subs(t=Pi/2,diff(z(t),t));

Per tant, ja podem calcular i dibuixar la recta i el pla tangent.
> r:=[Pi/2+t*v1,0+t*v2,Pi/2+t*v3];evalf(%);
> pla:=(x-Pi/2)*v1+(y-0)*v2+(z-Pi/2)*v3=0;evalf(%);
> dib1:=spacecurve([x(t),y(t),z(t)],t=0..2*Pi, axes=normal):
> dib2:=spacecurve([Pi/2+t*v1,0+t*v2,Pi/2+t*v3],t=-2..2,axes=normal):
> dib3:=implicitplot3d((x-Pi/2)*v1+(y-0)*v2+(z-Pi/2)*v3=0,x=-3..3,y=-3..3,z=-3..3,
axes=normal):
> display(dib1,dib2,dib3);

1.4 Estudi del comportament gràfic d’una funció

L’estudi de la primera i la segona derivada d’una funció ens ajuda a determinar el seu creixement i de-
creixement aix́ı com els intervals de concavitat i convexitat. Anem a fer un exemple i estudiarem la funció

f(x) =
5x

x2 − 4
.

Primer definim la funció i les seves derivades,
> restart;
> f:=x->(5*x)/(x^2-4);
> fp:=diff(f(x),x);
> fs:=diff(f(x),x$2);

Per estudiar el comportament a l’infinit, podem estudiar els limits quan x tendeixi a més i menys in-
finit. El Maple conté una comanda per calcular limits de funcions quan x tendeix a un valor o infinit: és
limit que té com a paràmetres la funció i el valor al qual tendeix x. Per exemple limit(f(x),x=a) calcula
el ĺımit de la funció quan x tendeix a a. A més podem especificar si el ĺımit el volem per la dreta o l’esquerre
afegint right o left com tercer paràmetre. Anem a estudiar primer els ĺımits a l’infinit.

> limit(f(x),x=infinity);
> limit(f(x),x=-infinity);

A més, hi ha dos punts x = 2,−2 que no pertanyen al domini de la funció (f(2) i f(−2) no estan definits)
i, per tant, cal estudiar el comportament de la funció quan x s’aproxima a aquests valors tant per la dreta
com per l’esquerra.

> limit(f(x),x=2,right); limit(f(x),x=2,left);
> limit(f(x),x=-2,right); limit(f(x),x=-2,left);

Per a determinar els extrems locals (f ′(x) = 0) i els intervals de creixement, analitzem la primera derivada.
> solve(fp=0);
> solve(fp>0);
> solve(fp<0);

Quan és la funció creixent?

Per a determinar els punts d’inflexió (f ′′(x) = 0) i els intervals de concavitat i convexitat analitzem la
segona derivada.

> solve(fs=0);
> solve(fs>0);
> solve(fs<0);

Com és la convexitat i concavitat de la funció?
Finalment, podem comprovar els càlculs realitzats i les conclusions obtingudes amb un dibuix de la funció.
> plot(f(x),x=-5..5,y=-7..7);

1.5 Polinomis de Taylor

El polinomi de Taylor es pot fer servir per aproximar una funció a l’entorn d’un punt a partir dels valors de
les seves derivades en aquest punt. Són els polinomis que més s’assemblen o millor aproximen a la funció i que
passen per un punt fixat. La comanda per a calcular directament les aproximacions de Taylor és taylor i
té com a paràmetres la funció, el punt que hem fixat i l’ordre del polinomi que volem obtenir, per exemple
taylor(f(x),x=a,n) on a és el punt on desenvolupem el polinomi i n és l’ordre de l’aproximació. Observeu
la seva forma genèrica en el següent exemple.

> restart;
> taylor(f(x),x=a,3);taylor(f(x),x=a,4);taylor(f(x),x=a,5);

La comanda per a trobar l’aproximació de la funció f(x) =
√

1 + x en el punt a = 0 fins a ordre 3 seria
taylor(sqrt(1+x),x=0,3);

> f:=x->sqrt(1+x);
> taylor(sqrt(1+x),x=0,3);

Anem a veure uns exemples més, i com els valors del polinomi de Taylor aproximen el valor de la funció.
Observeu que el resultat que dóna la comanda taylor no és una expressió. Per convertir-la a una expressió
polinomial cal utilitzar la comanda convert amb l’opció polynom (aquesta és una comanda molt general
que ja hav́ıem utilitzat per passar de graus a radians, vegeu l’ajuda de Maple)

> taylor(f(x),x=0,4);convert(%,polynom);
> subs(x=0.1,%);evalf(%);
> taylor(f(x),x=0,10);convert(%,polynom);
> subs(x=0.1,%);evalf(%);

El valor de la funció és
> evalf(f(0.1));

També podem representar gràficament les diferents aproximacions de Taylor per observar com van apro-
ximant la funció original cada cop amb més precisió a prop del punt que hem fixat,

> plot(f(x),x=-2..2,y=-3..3);
> p1:=convert(taylor(f(x),x=0,1),polynom);
> p2:=convert(taylor(f(x),x=0,2),polynom);
> p3:=convert(taylor(f(x),x=0,3),polynom);
> p4:=convert(taylor(f(x),x=0,4),polynom);
> p6:=convert(taylor(f(x),x=0,6),polynom);
> plot([f(x),p1,p2,p3,p4,p6],x=-2..2,y=-3..3);

Exercicis

1. Troba l’aproximació de Taylor de les funcions següents, en els punts i ordres que s’indiquen.

(a) f(x) = xex, a = 0, O = 3, 4, 5, 6.

(b) f(x) = x1/3, a = −8, O = 2, 3, 4;

2. Trobeu el polinomi de Taylor de grau 3 de la funció f(x) = ln(1 + x) en el punt a = 0. Quin és l’error
que es comet a l’usar aquest polinomi per aproximar el valor de ln(1.1)?

1.6 Càlcul de primitives

Una altra eina fonamental és el càlcul de primitives (o integració indefinida) juntament amb la integració
definida que permet calcular l’àrea de regions delimitades per gràfiques de funcions. Algunes de les comandes
que utilitzarem es troben al paquet student i Student[Calculus1] .

1.6.1 La comanda bàsica: int

La comanda bàsica pel càlcul de primitives o integrals indefinides és int . Té dos paràmetres: l’expressió o
funció f(x) que volem integrar i la variable respecte la qual integrem, és a dir, en la forma int(f(x),x).
Observeu que integra expressions i el resultat és una altra expressió. Comencem veient un exemple senzill,

> restart;
> f:=x->cos(x);
> int(f(x),x);

Per comprovar que el resultat és correcte només hem de derivar el resultat obtingut per comprovar que
és una primitiva de f(x),

> diff(%,x);

Una altre comanda interessant és Int . Té la mateixa sintaxi que int i és una comanda inert, és a dir,
ens escriu la integral per pantalla però no la calcula.

> Int(f(x),x);

Finalment, per executar una comanda inert existeix la comanda value . Aix́ı obtenim el resultat de
l’exemple,

> value(%);

Observeu doncs que la combinació value(Int) equival a int.
> int(x^2+x,x);

> Int(x^2+x,x);

> value(%);

> Int(1/(1+x^2),x);

> value(%);

> Int(1/sqrt(1-x^2),x);

> value(%);

1.6.2 Mètodes d’integració

En aquesta secció repassarem alguns dels mètodes clàssics del càlcul de primitives o càlcul integral. Alguns
d’aquest mètodes es troben implementats en comandes de manera que li podem dir al Maple que resolgui
una integral aplicant un mètode concret. Aquest tipus de comandes es troben al paquet student .

> restart; with(student);

Integració per substitució

El mètode de substitució es basa en la regla de la cadena i és efectiu quan volem calcular integrals expres-
sables en la forma

∫
f(g(x)) ( ∂

∂x g(x)) dx . Aleshores fem la substitució g(x) = u i per tant ∂
∂x g(x)dx = du que

converteix la anterior integral en
∫

f(u) du. Si hem fet el canvi adequat, aquesta serà més fàcil de calcular.
Recordeu que al final hem de tornar a desfer el canvi.

Per fer un canvi de variable amb Maple tenim la comanda changevar que té com a paràmetres el canvi
a realitzar, la integral i finalment la nova variable. Per exemple, changevar(g(x) = u, I, u) on I és la integral.

Observeu el següent exemple, on anem a calcular la integral indefinida de ex−2 e(2 x)

1−ex .
> f:=Int((exp(x)-2*exp(2*x))/(1-exp(x)),x);

Farem el canvi de variable ex = t.
> changevar(exp(x)=t,f,t);

> f:=simplify(%);

Aquesta nova expressió és més senzilla de calcular ja que és una funció racional, de fet, les funcions
racionals es poden descomposar en sumes de fraccions més senzilles amb la comanda convert que ja hem
utilitzat vàries vegades però ara amb uns paràmetres diferents, convert( fraccio ,parfrac, variable)
on el primer paràmetre és la fracció.

> convert((-1+2*t)/(-1+t),parfrac,t);

> value(f);

Finalment tornem a desfer el canvi.
> subs(t=exp(x),%);

Exercicis:

1. Calculeu la integral indefinida
∫

3 x2+1√
(1−x−x3) (1+x+x3)

dx. Quin és el canvi de variable més adient?

2. Simplifiqueu al màxim mitjançant un canvi de variable la integral
∫ x+1+2

√
x

x−
√

x
dx .

Integració per parts

El mètode d’integració per parts es desprén de la regla de la derivació del producte de funcions. Usualment
s’escriu

∫
u dv = u v −

∫
v du. Una vegada optem per aquest mètode el problema és decidir quina funció cal

prendre com a u.
La comanda de Maple per integrar per parts és intparts que té com a paràmetres la integral i la funció

que prenem com a u.
Observeu el següent exemple,
> restart; with(student);

Anem a descomposar la següent integral pel mètode d’ integració per parts de manera que la obtinguda
sigui més senzilla.

> f:=Int(x*exp(2*x),x);

Escollim u = x.
> intparts(f,x);

Fixeu-vos que la integral que resulta és més senzilla que la original. El resultat final és
> value(%);

Exercici:

Mitjançant el mètode de substitució o integració per parts simplifiqueu i calculeu les següents integrals
indefinides:

1.
∫

x2 ln(x) dx.

2.
∫

x arctan(x) dx.

3.
∫

arctan(x) dx.

Integració de funcions trigonomètriques

Per resoldre integrals trigonomètriques podem utilitzar alguns canvis especials. Per exemple, volem calcular
la integral indefinida de cos(x)5.

> restart; with(student);

> i:=Int(cos(x)^5,x);

Recordeu la relació trigonomètrica cos(x)2 + sin(x)2 = 1, i utilitzeu-la per canviar cos(x)4 = (1− sin(x)2)2.

> i:=Int((1-sin(x)^2)^2*cos(x),x);

Observeu com el canvi de variable u = sin(x) la converteix en una funció polinòmica fàcil d’integrar.

> changevar(u=sin(x),i,u);

> expand(%);

> value(%);

> subs(u=sin(x),%);

Exercici

Observant l’exercici anterior, calculeu
∫

cos(x)3 sin(x)2 dx.

Integració de funcions racionals

El càlcul de primitives per funcions racionals es basa en la descomposició de les fraccions en fraccions més
elementals per les quals podem calcular les primitives. Ja hem vist en un exemple anterior el funcionament de
la comanda convert per treballar amb fraccions de polinomis convert(fraccio, parfrac, variable).
Anem a veure uns exemples,

> restart; with(student):

> f:=x->(x+1)/(x^4-2*x^2+1);

> convert(f(x),parfrac,x);

> int(f(x),x);

> g:=x->36/(x^5-2*x^4-2*x^3+4*x^2+x-2);

> convert(g(x),parfrac,x);

> int(g(x),x);

> Int(1/(1+4*x^2),x);

> value(%);

1.6.3 Com obtenir una pista: la comanda Hint

El paquet Student[Calculus1] conté la funció Hint que t’ajudarà a saber com fer la primitiva.

> with(Student[Calculus1]):
> Hint(Int(x*exp(x),x));
> Hint(Int(cos(x)^5,x));

Mira l’ajuda que et dóna quan li demanes una pista sobre com calcular algunes de les primitives de les
seccions anteriors. Observa també les comandes que apareixen quan carregues el paquet.



Tema 3: Equacions diferencials 9

1.6.4 La integral definida: càlcul d’àrees

La integral definida d’una funció f(x) entre x = a i x = b és l’area (amb signe) de la regió delimitada per la
funció, l’eix de les x i les rectes verticals x = a i x = b.

> restart; with(student);

Prenem com a exemple la funció sinus entre 0 i π.
> f:=x->sin(x);

> plot(f(x),x=0..Pi, filled=true, color=red);

Aproximem la regió de color blau prenent rectangles d’una base de llargada fixa i fem la suma de l’àrea
d’aquests rectangles. El Maple al seu paquet student conté les comandes leftbox i rightbox que tenen
com a paràmetres una funció, un interval i el número de rectangles en que volem dividir la regió (prenent
l’alçada del rectangle el valor a l’esquerra o dreta de l’interval). El resultat és la representació gràfica. Les
comandes leftsum i rightsum ens donen la suma de les àrees dels rectangles.

> leftbox(f(x),x=0..Pi,20);

> leftsum(f(x),x=0..Pi,20);

> evalf(%);

> leftbox(f(x),x=0..Pi,40);

> leftsum(f(x),x=0..Pi,40);

> evalf(%);

Observeu que com més rectangles dibuixem, l’àrea obtinguda s’assembla més a l’àrea de la regió entre
l’eix horitzontal i la gràfica de la funció.

La integral definida és el ĺımit d’aquesta suma quan el número de rectangles en què subdividim la regió
tendeix a infinit. La comanda Limit és una comanda inert que només ens mostra la fórmula del ĺımit per
pantalla. Per executar-la i obtenir el resultat del ĺımit ha d’anar acompanyada amb la comanda value.

> Limit(leftsum(f(x),x=0..Pi,n),n=infinity);
> value(%);
> limit(leftsum(f(x),x=0..Pi,n),n=infinity);

La comanda int ens permet calcular integrals definides quan introduim l’interval d’integració, per exemple,
en el cas anterior faŕıem:

> Int(f(x),x=0..Pi);

> int(f(x),x=0..Pi);

Recordeu però què la integral definida calcula l’àrea amb signe, és a dir, la regió que queda per sota de
l’eix compta com a negativa

> leftbox(f(x),x=0..2*Pi,20);

> leftsum(f(x),x=0..2*Pi,20);

> value(%);

> Int(f(x),x=0..2*Pi);

> value(%);

En aquest cas l’àrea real és ( abs ens dóna el valor absolut d’un número).
> abs(Int(f(x),x=0..Pi))+abs(Int(f(x),x=Pi..2*Pi));

> value(%);

Exercici

Calculeu l’àrea de la regió delimitada entre les funcions f(x) = 5− x2 i g(x) = 3− x?

2 Equacions Diferencials

Una equació diferencial per a una funció y(t) és una equació en què la incògnita és una funció real y(t) i
l’expressió és del tipus F (t, y, y′, .., y(n)) = 0 on F és una funció de la variable independent t, la funció y(t) i
les derivades de y(t). És a dir, busquem una funció que compleixi una certa relació amb les seves derivades.
Aleshores, una solució és una funció y(t) de forma que quan substituim y per y(t), y′ per y′(t), etc, en
l’equació F (t, y, y′, .., y(n)) = 0 obtenim 0 idènticament en t.

2.1 Equacions diferencials elementals

Per exemple, la següent expressió és una equació diferencial,

y′(t) = cos(t)

on la incògnita és una funció y(t) que satisfà que la seva derivada és cos(t). Aquesta ja la sabeu resoldre,
només cal trobar una primitiva de la funció cos(t). És a dir, y(t) = sin(t)+C on C és un número real qualsevol.
Observeu que hi ha infinites solucions, una per a cada valor real de la constant C. Les equacions diferencials
del tipus y′(t) = f(t) es diuen elementals i les solucions venen donades per les funcions primitives de f(t).

2.2 Equacions diferencials més complicades: la comanda dsolve

Ara bé, podem trobar equacions més complicades, com per exemple,

y′(t) = y(t)

on el que ens demanen és que trobem funcions y(t) tals que siguin iguals a la seva derivada. En aquest cas, ja
en sabem una, et. Observeu que si y(t) = et aleshores es compleix que (et)′ = et i per tant, y′(t) = et = y(t).

Un altre exemple podria ser l’equació y′(t) = ty(t). En aquest cas anem a veure com usar Maple per
resoldre-la.

El Maple conté una comanda que ens permet trobar solucions d’equacions diferencials: és la instrucció
dsolve que permet resoldre una equació diferencial. Per trobar la solució de l’equació diferencial y′(t) =
ty(t) introdüım primer l’equació diferencial que anomenarem edo1.

> edo1:=diff(y(t),t)=t*y(t);

És molt important quan introdüım una equació diferencial en Maple no oblidar la variable independent
t, és a dir, sempre hem d’escriure y(t) perquè si només introdüım y el programa no ho entendrà i es pensarà
que és una constant.

Per a trobar la solució d’aquesta equació diferencial ara fem servir la instrucció dsolve

> dsolve(edo1,y(t));

obtenint la solució y(t) = Ce
t2

2 . Podem comprovar que és solució canviant la solució a l’equació, és a dir,
amb la solució concreta calculem y′(t) i ty(t) i veiem que coincideixen.

> subs(%,edo1); eval(%);

Noteu a més que hi ha infinites solucions, una per a cada valor de la constant C. Si afegim una condició
inicial, és a dir, donem el valor de la solució que busquem sobre un punt t, per exemple, demanem que
la solució y(t) compleixi que y(1) = 2 aleshores el que volem resoldre és el següent problema, que es diu
Problema de Valor inicial

y′ = ty
y(1) = 2

És a dir, per a quin valor de C es compleix que y(1) = 2, és a dir, Ce1/2 = 2. En el següent bloc de comandes,
tenim la instrucció rhs que extreu la part dreta d’una igualtat (la comanda lhs fa el mateix amb la part
esquerra). Substituim t = 1 a la solució, igualem a 2 i resolem l’equació per trobar el valor de la constant C.

> dsolve(edo1,y(t));
> subs(t=1,%);
> solve(rhs(%)=2);

Ara bé, la comanda dsolve ens permet trobar la solució d’aquest problema de valor inicial directament
i observeu que ara la solució śı que és única.

> dsolve({edo1,y(1)=2},y(t));
Donada una equació diferencial podem comprovar si una determinada funció n’és solució. Senzillament

mirant que les derivades compleixen la relació expressada en l’equació diferencial.

Exemple 1

Per exemple, donada l’equació diferencial y′ − y = 0, comprovem si la corba y(t) = et és solució.

> edo2:=diff(y(t),t)-y(t)=0;

> subs(y(t)=exp(t),edo2);

Avaluem l’expressió anterior

> %;

Observeu que obteniu 0 = 0, això vol dir que y(t) = et compleix y′(t)−y(t) = 0. Ara calculeu les solucions
de l’equació anterior amb la comanda dsolve i comproveu que us surt la solució que us hem donat.

> dsolve(edo2,y(t));

Exemple 2

Un altre exemple que analitzarem és el de l’equació diferencial y′ + 2y = et i comprovarem si la corba
y(t) = 2e−2t + 1

3et és solució o no.

> edo3:=diff(y(t),t)+2*y(t)=exp(t);

> subs(y(t)=2*exp(-2*t)+(1/3)*exp(t),edo3);

> %;

Ara calcularem totes les solucions,

> dsolve(edo3,y(t));

2.3 Equacions en variables separades

És molt útil al treballar amb equacions diferencials de primer ordre (només apareixen derivades de com a

molt primer ordre) escriure la funció derivada y′(t) com
dy

dt
.

Les equacions de variables separades són aquelles que es poden escriure sota la forma A(t)dt = B(y)dy.
Si les tenim escrites com dy

dt = f(t, y) és el cas en que f(t, y) és producte d’una funció de t per una funció

de y, en aquest cas,
dy

dt
= A(t)

1
B(y)

. La solució s’obté calculant primitives en ambdós costats de l’equació

A(t)dt = B(y)dy i äıllant y. L’exemple següent us ajudarà a entendre-ho millor.

Exemple:

L’equació y′ = t2y és de variables separades. Observeu que si fem el canvi
dy

dt
= t2y, i separem tot el que té

t del que té y tenim que
1
y
dy = t2dt. Aleshores només cal integrar ambdós costats i äıllar y en el resultat.

En aquest cas tindŕıem
> int(1/y,y)=int(t^2,t)+C;
> y(t)=solve(%,y);

Fent servir el maple la podem solucionar utilitzant la comanda dsolve i obtenir totes les possibles
solucions.

> dsolve(diff(y(t),t)=t^(2)*y(t),y(t));

Exercici

1. Comproveu que les següents equacions són de variables separades i trobeu la solució general.

a) y′ = (y − 1)t,

b) y′ = et(y3 − 1).

c) y′ = yt2 + y sin(t).

2. Trobeu la solució general de l’equació diferencial i la que compleix la condició inicial

a) y′ = et(y − 1), y(0) = e.

b) y′ = et(y2 − y), y(0) = 1
2 .

c) y′ = x ln(y), y(0) = 1.

2.4 Equacions lineals de primer ordre

Són les equacions diferencials del tipus y′(t) − a(t)y(t) = b(t) on a(t) i b(t) són dues funcions donades que
només depenen de t. La solució general d’aquest tipus d’equacions y′(t) − a(t)y(t) = b(t) es descomposa en
una suma de la forma y(t) = yp(t) + yh(t) on yp(t) és una solució particular qualsevol i yh(t) és la solució de
l’equació homogènia y′(t)− a(t)y(t) = 0 (és a dir, fent b(t) = 0).

Noteu que l’equació homogènia y′(t) − a(t)y(t) = 0 és de variables separades: la podem escriure com
dy

dt
=

a(t)y, és a dir,
1
y
dy = a(t)dt. I per tant, ln(y) =

∫
a(t)dt + C. És a dir, y(t) = Ce

R
a(t)dt.

Exemple:

Anem a estudiar l’equació t3y′ + 2ty = e
2
t que és una equació diferencial lineal ja que la podem escriure com

y′ + 2t
t3

y = e2/t

t3
. Si la resolem fent servir la instrucció dsolve

> dsolve(t^3*diff(y(t),t)+2*t*y(t)=exp(2/t),y(t));

obtenim que la solució és y(t) =
(
− 1

2t2
+ c

)
e

2
t ,

Anem a veure en el següent exercici com una part d’aquesta solució (la que té la constant) correspont a
la solució de l’equació diferencial homogènia associada t3y′ + 2ty = 0 i l’altra part que queda sumant és una
solució de la no homogènia.

Exercici:

1. Anem a estudiar l’equació homogènia. Comproveu que yh(t) = ce
2
t és solució de l’equació homogènia

t3y′ + 2ty = 0. Ara comproveu que yp(t) =
(
− 1

2t2

)
e

2
t (la part de la solució general sense constant o la

que obtenim fent C = 0) és una solució particular de t3y′ + 2ty = e
2
t Observeu que la solució general

que hav́ıeu obtingut és la suma
(
− 1

2t2

)
e

2
t + ce

2
t .

2. Quina és la solució general de les equacions lineals següents:

(a) 4ty′ − 3y = t4.

(b) y′ + y = 2te−t + t2.

(c) 2y − t3 = ty′.

Per a cadascuna d’elles doneu la solució de l’equació homogènia corresponent i una solució particular.

2.5 Equacions diferencials lineals a coeficients constants

Una equació diferencial lineal amb coeficients constants és una equació del tipus any(n) + an−1y
(n−1) + . . . +

a1y
′ +a0y = f(t), on ai són números reals. Com en el cas anterior, les seves solucions també es descomponen

en una suma de la forma y(t) = yp(t)+yh(t) on yp(t) és una solució particular i yh(t) és la solució de l’equació
homogènia any(n) + an−1y

(n−1) + . . . + a1y
′ + a0y = 0.

L’equació homogènia associada

L’equació homogènia any(n) +an−1y
(n−1) + . . .+a1y

′ +a0y = 0. Es resol a partir de les solucions de l’equació
caracteŕıstica que s’obté canviant y(i) per λi a l’equació diferencial, és a dir, Pn(λ) = anλn +an−1λ

n−1 + . . .+
a1λ+ a0 = 0. Aleshores les arrels d’aquest polinomi ens permeten calcular les solucions. Si λi és una arrel de
l’equació caracteŕıstica llavors cie

λit és solució de l’equació homogènea anx(n)+an−1x
(n−1)+. . .+a1x

′+a0x =
0.

Anem a veure la relació amb uns exemples.

Exemple 1

Considerem l’equació y′′ − 3y′ + 2y = 0. En aquest cas l’equació caracteŕıstica és P2(λ) = λ2 − 3λ + 2 = 0.
Les seves arrels són 2 i 1.

Trobem ara la seva solució amb la comanda dsolve

> edo:=diff(y(t),t$2)-3*diff(y(t),t)+2*y(t)=0;

> dsolve(edo,y(t));

La solució és
y(t) = c1e

2t + c2e
t.

Observeu que per cada una de les arrels anteriors 2 i 1 obtenim una solució de la forma exponencial e1·t i e2t.
A més, qualsevol combinació lineal d’aquestes és solució. Per això diem que {et, e2t} formen una base de la
solució homogènia.

Sempre que el polinomi caracteŕıstic tingui solucions reals amb multiplicitat 1 passarà el mateix, és a dir,
per cada arrel λi tindrem una solució eλit. I totes elles formen una base de solucions de manera que totes les
seves combinacions lineals també ho són.

Exemple 2

Considerem l’equació y′′ + 2y′ + y = 0. En aquest cas l’equació caracteŕıstica és P2(λ) = λ2 + 2λ + 1 = 0.
Les seves arrels són −1 i −1. És a dir, −1 amb multiplicitat 2.

Trobem ara la seva solució amb la comanda dsolve

> edo:=diff(y(t),t$2)+2*diff(y(t),t)+y(t)=0;

> dsolve(edo,y(t));

La solució és
y(t) = c1te

−t + c2e
−t.

Observeu que com que −1 és arrel obtenim una solució de la forma exponencial e−1·t. Però com que la
seva multiplicitat és 2, també apareix una nova solució de la forma te−t. A més, qualsevol combinació lineal
d’aquestes és solució. Per això diem que {e−t, te−t} formen una base de la solució homogènia.

Sempre que el polinomi caracteŕıstic tingui solucions reals amb multiplicitat m passarà el mateix, és a
dir, per cada arrel λi tindrem solucions tieλit per i = 0, . . . ,m− 1. I totes elles formen una base de solucions
de manera que totes les seves combinacions lineals també ho són. Feu el següent exercici per entendre millor
aquesta relació.

Exercici

Estudieu la relació entre les arrels del polinomi caracteŕıstic i les solucions de l’equació diferencial per la
següent equació y(iv) + 3y

′′′ − 6y
′′ − 28y′ − 24y = 0.

Exemple 3

Considerem l’equació y′′ + 2y′ + 5y = 0. En aquest cas l’equació caracteŕıstica és P2(λ) = λ2 + 2λ + 5 = 0.
Les seves arrels són complexes −1 + 2i i −1− 2i.

Trobem ara la seva solució amb la comanda dsolve

> edo:=diff(y(t),t$2)+2*diff(y(t),t)+5*y(t)=0;

> dsolve(edo,y(t));

La solució és
y(t) = +c1e

−t sin(2t) + c2e
−t cos(2t).

Observeu que com que −1 + 2i i la seva conjugada −1 − 2i són arrels obtenim una solució de la forma
exponencial per trigonomètrica e−1·t cos(2t) i e−1·t sin(2t). A més, qualsevol combinació lineal d’aquestes és
solució. Per això diem que {e−1·t cos(2t), e−1·t sin(2t)} formen una base de la solució homogènia.

Sempre que el polinomi caracteŕıstic tingui solucions complexes a + bi i a− bi passarà el mateix, és a dir,
per cada parella d’arrels complexes conjugades a+ bi i a− bi tindrem solucions eat cos(bt) i eat sin(bt). I totes
elles formen una base de solucions de manera que totes les seves combinacions lineals també ho són.

Exercicis

1. Resoleu les equacions d’ordre superior següents:

(a) y′′ + y′ − 6y = 0.

(b) y′′′ − 2y′′ − 3y′ = 0.

(c) y′′ + 2y′ + y = 0.

(d) y(v) − 2y(iv) + 2y′′′ − 4y′′ + y′ − 2y = 0.

Doneu el polinomi caracteŕıstic i les seves arrels i estudieu la relació amb les solucions.

2. Considereu els problemes de valor inicial següents:

(a) y′′′ − y′ = −2t, y(0) = 0, y′(0) = 1, y′′(0) = 2.
> edo:=diff(y(t),t$3)-diff(y(t),t)=-2*t;
> dsolve({edo,y(0)=0,D(y)(0)=1,D(D(y))(0)=2},y(t));

(b) y′′ + y = 2 cos t, y(0) = 1, y′(0) = 0.

Escriviu la solució general de l’equació homogènia, una solució particular de l’equació no homogènia i
la única solució que compleix totes les condicions inicials.

2.6 Equacions diferencials d’ordre 2. Forçament i ressonància

En el fitxer ressonancia.mws es mostren les solucions de l’equació y′′ + 2y = cos(ωt), amb condicions
y(0) = 0, y′(0) = 0 per a diversos valors de ω. El terme periòdic cos(ωt) s’anomena forçament. El polinomi
caracteŕıstic de equació és λ2 + 2, observeu que les seves arrels són ±

√
2i. Fixeu-vos ara que per a ω 6= ±

√
2

les solucions no són periòdiques però śı acotades, mentre que per a ω = ±
√

2 no són ni periòdiques ni
acotades, és a dir, que l’amplitud de les solucions creix amb el temps. Aquest fenomen s’anomena resonància
i és produeix quan la frequència de la força externa (en aquest cas

√
2) coincideix amb la frequència interna

del sistema (la solució de l’equació caracteŕıstica).

Exercicis:

1. Considereu l’equació y′′ + 2y′ + 2y = cos(ωt), amb condicions y(0) = 0, y′(0) = 0. Trobeu la solució i
dibuixeu-la per als valors ω = 1, ω = 3, ω = 7

5 , ω =
√

2. Hi ha algun valor de ω pel qual es produeixi
ressonància?

2. Considereu l’equació y′′ + 4y = cos(at) amb condicions inicials y(0) = 1, y′(0) = 2.

a) Doneu la solució per a = 1. Està acotada?

b) Doneu la solució per a = 4. Està acotada?

c) Doneu la solució per a qualsevol. Per a quin valors del paràmetre a la solució no està acotada?
Doneu la solució per aquests valors del paràmetre a.


